Numerical Calculation of Lyapunov Exponents

Marco Sandri, University of Verona, Italy

The Lyapunov characteristic exponents play a crucial role in the description of the behavior of dynamical systems. They measure the average rate of divergence or convergence of orbits starting from nearby initial points. Therefore, they can be used to analyze the stability of limits sets and to check sensitive dependence on initial conditions, that is, the presence of chaotic attractors. This article shows how to use *Mathematica* to compute the Lyapunov spectrum of a smooth dynamical system.

A large number of fields of human knowledge, from physics to chemistry, from medicine to meteorology, from biology to economics, makes use of difference and differential equations (linear or nonlinear, ordinary or partial) to describe and explain facts observed in the real world (see, for example, [Beltrami 1989]). Advances made in the last two decades in the analysis of nonlinear systems have led to new areas of application and research for dynamical system theory. Among these developments is the discovery that deterministic systems can exhibit a wide range of erratic behaviors (usually called "chaos"), in many respects indistinguishable from noise. Several approaches have developed for the study of such systems, both geometrical [Guckenheimer and Holmes 1983] and statistical [Eckmann and Ruelle 1985]. The statistical approach, which is rooted in ergodic theory, seeks to characterize dynamical systems through concepts such as entropy, dimension, and Lyapunov characteristic exponents. Roughly speaking, dimensions express the number of excited degrees of freedom of the system and the degree of "geometric complexity" of the attractor, while entropies describes the production of information in the system. Lyapunov characteristic exponents (LCEs) measure the separation in time of two orbits starting from arbitrary close initial points. All these quantities are used to quantify the erratic or chaotic behavior of a system's dynamics.

In this article, we develop a package for estimating the Lyapunov exponents of continuous and discrete differentiable dynamical systems, based on the algorithms presented in [Benettin et al. 1980] and [Eckmann and Ruelle 1985].

Differentiable Dynamical Systems

In this section, we briefly describe some basic notions of dynamical system theory that will be useful in the next sections. For a more detailed treatment, see [Guckenheimer and Holmes 1983].

Marco Sandri studied economics at the Faculty of Economics and Commerce of the University of Verona, Italy, and is now working at the Calculus Center of that university as manager of the Scientific Calculus Area. His research interests include time series analysis, neural networks, wavelets, and dynamical system theory.

An n-dimensional *continuous-time* (autonomous) smooth dynamical system is defined by the differential equation

$$\dot{x} = F(x), \tag{1}$$

where $\dot{x} = dx/dt$, $x(t) \in \mathbf{R}^n$ is the state vector at time t and $F : U \to \mathbf{R}^n$ is a C^r function ($r \geq 1$) on an open set $U \subset \mathbf{R}^n$ (that is, F has derivatives of order r which are continuous at each point of U). Here, we will not explicitly consider nonautonomous systems $\dot{x} = F(x, t)$, because, for our purposes, it will be sufficient to treat t as an additional dependent variable with the trivial evolution equation $\dot{t} = 1$. In other words, we will rewrite every nonautonomous system as an autonomous system $\dot{x} = F(x, t)$, $\dot{t} = 1$. (at the expense of increasing the dimension by one).

The space of dependent variables is often referred to as the phase space M of the system, which, in our discussion, will be $\mathbf{R}^n$. The function F is a called *vector field*. If it is linear, then the system (1) is linear. Moreover, we say that the vector field F generates the *flow* $f : U \times \mathbf{R} \to \mathbf{R}^n$, where $f^t(x) = f(x, t)$ is a C^r function defined for all $x \in U$ and $t \in \mathbf{R}$, such that

$$\dot{f}^t(x) = F(f^t(x)) \quad \text{for all } x \in U, t \in \mathbf{R}.$$

Given an initial state $x_0 \in U$, the *solution* of (1) is the function $f^t(x_0) : \mathbf{R} \to \mathbf{R}^n$ such that $f^0(x_0) = x_0$. The set $\{f^t(x_0) : t \in \mathbf{R}\}$ is called the *trajectory* of the system through x_0.

Any C^r map $F : U \to \mathbf{R}^n$ on an open set $U \subset \mathbf{R}^n$ defines an n-dimensional *discrete-time* (autonomous) smooth dynamical system by the state equation

$$x_{t+1} = f(x_t), \; t \in \mathbf{N},$$

where $x_t \in \mathbf{R}^n$ is the state of the system at time t and f maps x_t to the next state x_{t+1}. Starting with an initial condition x_0, repeated applications (iterates) of f generate a discrete set of points (the *orbit*) $\{f^t(x_0) : t \in \mathbf{N}\}$, where $f^t(x) = f \circ \cdots \circ f(x)$.

One of the main goals of dynamical system theory is the study of steady-state behaviors, that is the quantitative and qualitative description of the asymptotic evolutions of systems as $t \to \infty$. An important concept for this purpose is the notion of a limit set.

A point p is an ω-limit point of x if there are points $f^{t_1}(x)$, $f^{t_2}(x)$, ... on the trajectory of x such that $f^{t_i}(x) \to p$ as $t_i \to \infty$. The *ω-limit set* $\Omega(x)$ is the set of all the ω-limit points of x. Moreover, an ω-limit set Ω is *attracting* if there exists an open neighborhood U of $\Omega(x)$ such that $\Omega(x) = \Omega$ for all $x \in U$. The *basin of attraction* B_Ω of an attracting set Ω is the union of all such neighborhoods U. In other words, B_Ω is the set of all initial conditions x that tend toward Ω as $t \to \infty$. (We make no distinction between attracting limit sets and attractors, even if they are quite different; from a practical point of view, there is little harm in ignoring the difference.)

Of course, for a scientist, only attracting limit sets are of interest because non-attracting limit sets cannot be observed in real systems and simulations. As we will show, LCEs are a useful tool to investigate the stability of limit sets and to classify systems on the basis of their asymptotic evolution.

A dynamical system may have only one attracting limit set, or it may have several, each one with a different basin of attraction. In this case, the initial condition determines which limit set will be approached. There are four foundamental types of limit sets, corresponding to as many types of solutions of differential (or difference) equations.

1. *Fixed points.* A fixed point is a point $x \in M$ such that $f^t(x) = x$ for all t.
2. *Periodic motions.* A periodic motion is a solution f^t such that $f^t(x) = f^{t+T}(x)$ for some fixed constant $T > 0$ (the period) and all t. The limit set corresponding to a periodic solution is the closed curve traced out by $f^t(x)$ over one period, which is topologically equivalent to a circle S^1.
3. *Quasiperiodic motions.* A quasiperiodic solution of a dynamical system is a function $f : \mathbf{R} \to \mathbf{R}^n$ that can be represented in the form $f(t) = H(\omega_1 t, \ldots, \omega_n t)$, where H is periodic of period 2π in each argument, and the real numbers $\omega_1, \ldots, \omega_n$ form a finite set of base frequencies (see [Parker and Chua 1989, 13–18]). A quasiperiodic solution with q base frequencies is called q-periodic. The limit set of a q-periodic solution is a diffeomorphic copy of a q-dimensional torus $\mathbf{T}^q = S^1 \times \cdots \times S^1$, where each S^1 represents one of the base frequencies.
4. *Chaotic motions.* Avoiding a formal definition, we say that chaotic dynamics are characterized by three properties: (*a*) they are bounded random-like steady-state trajectories distinct from the previous kinds of motion; (*b*) they converge to a set in the phase space, called a *strange attractor*, which is not a simple manifold like a point, circle, or torus, but has a complex (fractal) geometrical structure with a fractional Hausdorff dimension [Falconer 1985]; (*c*) they exhibit sensitive dependence to initial conditions, that is, chaotic trajectories locally diverge away from each other and small changes in starting conditions build up exponentially fast into large changes in evolution.

Lyapunov Exponents

Lyapunov exponents provide a quantitative measure of the divergence or convergence of nearby trajectories for a dynamical system. If we consider a small hypersphere of initial conditions in the phase space, for sufficiently short time scales, the effect of the dynamics will be to distort this set into a hyperellipsoid, stretched along some directions and contracted along others. The asymptotic rate of expansion of the largest axis, corresponding to the most unstable direction of the flow, is measured by the largest LCE λ_1. In general, if we sort the axes and LCEs in decreasing order by magnitude ($\varepsilon_1 \geq \cdots \geq \varepsilon_n$ and $\lambda_1 \geq \cdots \geq \lambda_n$), each λ_i quantifies the average exponential rate of expansion or contraction for the i-th axis ε_i.

More formally, consider two nearby points, x_0 and $x_0 + u_0$, in the phase space M, where u_0 is a small perturbation of the initial point x_0 (see Figure 1). After a time t, their images under the flow will be $f^t(x_0)$ and $f^t(x_0 + u_0)$ and the perturbation u_t will become

$$u_t \equiv f^t(x_0 + u_0) - f^t(x_0) = D_{x_0} f^t(x_0) \cdot u_0, \tag{2}$$

where the last term is obtained by linearizing f^t. Therefore the average exponential rate of divergence or convergence of the two trajectories is defined by

$$\lambda(x_0, u_0) = \lim_{t\to\infty} \frac{1}{t} \ln \frac{\|u_t\|}{\|u_0\|} = \lim_{t\to\infty} \frac{1}{t} \ln \|D_{x_0} f^t(x_0) \cdot u_0\|, \tag{3}$$

where $\|u\|$ denotes the length of a vector u. If $\lambda(x, u) > 0$, then one has exponential divergence of nearby orbits. It can be shown that, under very weak smoothness conditions on the dynamical system, the limit (3) exists and is finite for almost all points $x_0 \in M$, and, for almost all tangent vectors u_0, it is equal to the largest LCE λ_1 [Oseledec 1968].

Definition (3) refers to LCEs of vectors, also called LCEs of order 1. A natural and, for our purposes, useful generalization is to define LCEs of order p, $1 \leq p \leq n$, which describe the mean rate of growth of a p-dimensional volume in the tangent space. Consider a parallelepiped U_0 in the tangent space whose edges are the p vectors $u_1, \ldots, u_p$. LCEs of order p are then defined by

$$\lambda^p(x_0, U_0) = \lim_{t\to\infty} \frac{1}{t} \ln \left[\mathrm{Vol}^p(D_{x_0} f^t(U_0))\right], \tag{4}$$

where Vol^p is the p-dimensional volume defined in the tangent space. One of the theorems given in [Oseledec 1968] shows that one can find p linearly independent vectors $u_1, \ldots, u_p$, such that

$$\lambda^p(x_0, U_0) = \lambda_1 + \cdots + \lambda_p. \tag{5}$$



FIGURE 1. Divergence of two orbits starting from nearby initial points.

That is, each LCE of order p is equal to the sum of the p largest LCEs of order 1. For $p = n$, we obtain the mean exponential rate of growth of the phase space volume. Thus, for a measure-preserving flow (such as a Hamiltonian system), $\sum_{i=1}^{n} \lambda(x, u_i) = 0$, while for an attractor of a dissipative system, contraction must outweigh expansion and the sum must be negative. Haken [1983] proved that in continuous-time systems, for any attractor other than a fixed point, one Lyapunov exponent is always zero.

As previously mentioned, LCEs are convenient for categorizing asymptotic behaviors of dynamical systems. Following [Klein and Baier 1991], in Table 1 we give a classification of (autonomous) continuous-time attractors on the basis of their Lyapunov spectrum, together with their Hausdorff dimension. A similar scheme for discrete-time attractors can be constructed, though one needs to make a distinction between invertible and non-invertible maps (see [Klein and Baier 1991, 8]).

Topological dimension	Dynamics of the attractor	LCE spectrum	Hausdorff dimension
1	Fixed point	$-$	0
2	Periodic motion	$0\,-$	1
3	Torus T^2	$0\,0\,-$	2
	Chaos C^1	$+\,0\,-$	$2 < D < 3$
4	Hypertorus T^3	$0\,0\,0\,-$	3
	Chaos on T^3	$+\,0\,0\,-$	$3 < D < 4$
	Hyperchaos C^2	$+\,+\,0\,-$	$3 < D < 4$
N	Fixed point	$- \cdots -$	0
	Periodic Motion	$0 - \cdots -$	1
	$(N-1)$-torus	$\underbrace{0 \cdots 0}_{l \geq 2}\underbrace{- \cdots -}_{N-l}$	l
	$(N-2)$-chaos	$\underbrace{+ \cdots +}_{k \geq 1}\underbrace{0 \cdots 0}_{l \geq 1}\underbrace{- \cdots -}_{N-k-l}$	$k + l < D < N$

TABLE 1. LCE spectrum of continuous-time attractors (from [Klein and Baier 1991]).

Estimation of LCEs of Continuous Systems

Let us consider the continuos-time dynamical system (1), together with an initial point x_0 lying in the basin of attraction of the limit set. It is easy to show that the tangent vector u_t defined in (2) evolves in time satisfying the so-called *variational equation* (see [Parker and Chua 1989, 305–306]):

$$\dot{\Phi}_t(x_0) = D_x\, F(f^t(x_0)) \cdot \Phi_t(x_0), \qquad \Phi_0(x_0) = I, \tag{6}$$

where $\Phi_t(x_0)$ is the derivative with respect to x_0 of f^t at x_0, that is, $\Phi_t(x_0) = D_{x_0}\, f^t(x_0)$. Equation 6 is a matrix-valued time-varying linear differential equation whose coefficients depend on the evolution of the original system (1). Therefore, to calculate the trajectory, we must integrate the combined system

$$\begin{Bmatrix} \dot{x} \\ \dot{\Phi} \end{Bmatrix} = \begin{Bmatrix} F(x) \\ D_x F(x) \cdot \Phi \end{Bmatrix}, \qquad \begin{Bmatrix} x(t_0) \\ \Phi(t_0) \end{Bmatrix} = \begin{Bmatrix} x_0 \\ I \end{Bmatrix}. \tag{7}$$

As an example, let us suppose that the system (1) is the Lorenz model,

```
In[1]:= F[{x_, y_, z_}] := {16(y - x), x(45.92 - z) - y, x y - 4z}
```

with the initial condition

```
In[2]:= x0 = {19, 20, 50};
```

We use Roman Maeder's function `RKStep` to implement the numerical integration of the first-order differential equation (7) by the Runge-Kutta method [Maeder 1991, 171–175].

```
In[3]:= RKStep[f_, y_, y0_, dt_] :=
        Module[{ k1, k2, k3, k4 },
            k1 = dt N[ f /. Thread[y -> y0] ];
            k2 = dt N[ f /. Thread[y -> y0 + k1/2] ];
            k3 = dt N[ f /. Thread[y -> y0 + k2/2] ];
            k4 = dt N[ f /. Thread[y -> y0 + k3] ];
            y0 + (k1 + 2 k2 + 2 k3 + k4)/6 ]
```

```
In[4]:= IntVarEq[F_List, DPhi_List, x_List, Phi_List,
                 x0_List, Phi0_List, {t1_, dt_}] :=
        Module[{n, f, y, y0, yt},
          n = Length[x0];
          f = Flatten[Join[F, DPhi]];
          y = Flatten[Join[x, Phi]];
          y0 = Flatten[Join[x0, Phi0]];
          yt = Nest[ RKStep[f, y, #, N[dt]]&,
               N[y0], Round[N[t1/dt]] ];
          {First[#], Rest[#]}&  @  Partition[yt, n] ]
```

We also define functions to compute the Jacobian matrix of F and the Euclidean norm of a vector.

```
In[5]:= JacobianMatrix[funs_List, vars_List] :=
          Outer[D, funs, vars]
```

```
In[6]:= Norm[x_] := Sqrt[x.x]
```

Choosing a step size of 0.02 and $T = 20$, we compute the 3×3 matrix $\Phi_T(x_0) = D_{x_0} f^T(x_0)$:

```
In[7]:= T = 20;

In[8]:= stepsize = 0.02;

In[9]:= n = Length[x0];

In[10]:= x = Array[a, n];

In[11]:= Phi = Array[b, {n, n}];

In[12]:= DPhi = Phi.Transpose[JacobianMatrix[F[x], x]];

In[13]:= Phi0 = IdentityMatrix[n];

In[14]:= {xT, PhiT} =
           IntVarEq[F[x], DPhi, x, Phi, x0, Phi0, {T, stepsize}];

In[15]:= xT

Out[15]= {17.7334, 15.016, 53.42}
```

In[16]:= PhiT // MatrixForm

Out[16]//MatrixForm=

```
3.16998 10^11    -4.95482 10^12    5.81422 10^12
5.39883 10^11    -8.43862 10^12    9.90228 10^12
2.64416 10^11    -4.13294 10^12    4.84979 10^12
```

In the previous section, we mentioned that $\lambda(x_0, u_0) = \lambda_1$ for almost all initial conditions x_0 and for almost all tangent vectors u_0. Therefore, we can easily obtain an estimate of the largest LCE by directly applying the definition (3):

```
In[17]:= u = Table[Random[], {n}];

In[18]:= Log[Norm[PhiT.u]]/T

Out[18]= 1.46574
```

The calculation of the entire LCE spectrum is more problematic. Oseledec [1968] shows that one can obtain the entire spectrum by computing the eigenvalues of the matrix $\Phi_T^*(x_0)\cdot\Phi_T(x_0)$, where $\Phi_T^*(x_0)$ is the adjoint matrix of $\Phi_T(x_0)$, and using the fact that they behave like $e^{2T\lambda_1}, \ldots, e^{2T\lambda_n}$. Unfortunately, for large T, $\Phi_T(x)$ is an ill-conditioned matrix, because its columns tend to line up with the eigenvector associated to the largest eigenvalue. Consequently, the small relative errors in the largest eigenvalue might contaminate the smaller ones, giving unreliable estimates.

To circumvent these problems, we adopt the algorithm discussed in [Benettin et al. 1980], which relies on the calculation of the order-p LCEs defined in equation 4 and the repeated application of the Gram-Schmidt orthonormalization procedure.

Given a set $\{u_1, \ldots, u_p\}$ of p linearly independent vectors in $\mathbf{R}^n$, the Gram-Schmidt procedure generates an orthonormal set $\{v_1, \ldots, v_p\}$ of vectors which spans the same subspace spanned by $\{u_1, \ldots, u_p\}$. The vectors v_i are given by

$$\begin{aligned} w_1 &= u_1, \qquad v_1 = w_1/\|w_1\| \\ w_2 &= u_2 - \langle u_2, v_1\rangle v_1, \qquad v_2 = w_2/\|w_2\|, \quad \ldots \qquad (8) \\ w_p &= u_p - \sum_{i=1}^{p-1} \langle u_p, v_i\rangle v_i, \quad v_p = w_p/\|w_p\|, \end{aligned}$$

where $\langle \cdot, \cdot \rangle$ is the scalar product of vectors. It is easy to show that the volume of the parallelepiped spanned by $\{u_1, \ldots, u_p\}$ is

$$\mathrm{Vol}\,\{u_1,\ldots,u_p\} = \|w_1\| \cdots \|w_p\|.$$

Following the algorithm of [Benettin et al. 1980], we start by choosing an initial condition x_0 and an $n \times n$ matrix $U_0 = [u_1^0, \ldots, u_n^0]$. Using the Gram-Schmidt procedure, we calculate the corresponding matrix of orthonormal vectors $V_0 = [v_1^0, \ldots, v_n^0]$ and integrate the variational equation (7) from $\{x_0, V_0\}$ for a short interval T, to obtain $x_1 = f^T(x_0)$ and

$$U_1 \equiv [u_1^1, \ldots, u_n^1] = D_{x_0} f^T(U_0) = \Phi_T(x_0) \cdot [u_1^0, \ldots, u_n^0].$$

Again, we calculate the orthonormalized version of U_1 and integrate the equation from $\{x_1, V_1\}$ for T seconds to obtain x_2 and U_2. We repeat this integration-orthonormalization procedure K times.

During the k-th step, the p-dimensional volume Vol^p defined in (4) increases by a factor of $\|w_1^k\| \cdots \|w_p^k\|$, where $\{w_1^k, \ldots, w_p^k\}$ is the set of orthogonal vectors calculated from U_k using (8). The definition (4) then implies

$$\lambda^p(x_0, U_0) = \lim_{k\to\infty} \frac{1}{kT} \sum_{i=1}^{k} \ln(\|w_1^i\| \cdots \|w_p^i\|).$$

Subtracting λ^{p-1} from λ^p and using (5), we obtain the p-th LCE of order 1:

$$\lambda_p = \lim_{k\to\infty} \frac{1}{kT} \sum_{i=1}^{k} \ln \|w_p^i\|.$$

This relation suggests an easy way to calculate the Lyapunov spectrum. For a suitable value of T, continue to calculate the quantities

$$\frac{1}{KT} \sum_{i=1}^{K} \ln \|w_1^i\| \approx \lambda_1, \quad \ldots, \quad \frac{1}{KT} \sum_{i=1}^{K} \ln \|w_n^i\| \approx \lambda_n,$$

until they show convergence (for example, using the relative/absolute convergence test proposed in [Parker and Chua 1989, 302]) or until a maximum (sufficiently large) iteration count is reached.

This procedure can be easily implemented with *Mathematica*. The function `GramSchmidt` is defined in the package ``LinearAlgebra`Orthogonalization` ``.

```
In[19]:= << LinearAlgebra`Orthogonalization`

In[20]:= T = 0.1;

In[21]:= PhiT = Phi0;

In[22]:= K = 800;

In[23]:= s = {};

In[24]:= Do[
           {xT, PhiT} = IntVarEq[F[x], DPhi, x,
                    Phi, xT, PhiT, {T, stepsize}];
           W = GramSchmidt[PhiT, Normalized -> False];
           norms = Map[Norm, W];
           s = Append[s, norms];
           PhiT = W/norms,
           {K} ];

In[25]:= lces = Rest[FoldList[Plus, 0, Log[s]]]/(T Range[K]);
```

The three LCEs are

```
In[26]:= Last[lces]

Out[26]= {1.48245, -0.0001195, -22.4654}
```

To show the convergence of the LCEs, we use the `ConvergencePlot` command defined in the package `LCE.m`:

```
In[27]:= << LCE.m
```

```
In[28]:= ConvergencePlot[lces]
```



FIGURE 2. Convergence plot of the Lyapunov spectrum for the Lorenz model.

We can obtain the same results using the `LCEsC` command:

```
In[29]:= LCEsC[F, {19, 20, 50}, 0.1, 800, 20, 0.02]
Out[29]= {{1.48245, -0.0001195, -22.4654}, 2.06598}
```

This function returns the LCE spectrum and the Lyapunov dimension of the attractor. The convergence plot of the LCEs can be displayed by setting the option `LCEsPlot` to `True`.

An interesting example of a hyperchaotic continuous-time attractor ("Rössler hyperchaos") with two positive LCEs is given by

```
In[30]:= rossler[{x_, y_, z_, w_}] :=
           {- y - z, x + 0.25 y + w, 3 + x z, 0.05 w - 0.5 z}
```

(See [Rössler 1991].) The LCE spectrum is

```
In[31]:= lcesrossler =
          LCEsC[rossler, {-10, -14, 0.3, 29}, 0.1, 2000, 1, 0.02]
Out[31]= {{0.142433, 0.005138, -0.00407495, -24.0831}, 3.00596}
```

A 3D projection of the four-dimensional attractor of this dynamical system can be obtained with the `PhaseSpaceC` command.

```
In[32]:= PhaseSpaceC[rossler,
           {-10, -14, 0.3, 29}, 100, 20, 0.02, {1,2,3}]
```



FIGURE 3. A 3D projection of the four-dimensional Rössler hyper-chaotic attractor.

The last argument `{1,2,3}` in the `PhaseSpaceC` command specifies that the first three variables of the model (`x`, `y`, and `z`) are to be plotted on the x, y, z axes, respectively.

Finally, we can also make use of the algorithm in [Benettin et al. 1980] for estimating the LCEs of a nonautonomous system, for example the Duffing model:

```
In[33]:= duffing[{x_, y_, t_}] := {y, -0.1 y - x^3 + 10 Cos[t], 1}
```

```
In[34]:= LCEsC[duffing, {0.5, 0.7, 0.2}, 0.05, 1000, 10, 0.02,
           LCEsPlot -> True]
Out[34]= {{0.0791481, 0., -0.159148}, 2.49732}
```



FIGURE 4. Convergence plot of the Lyapunov spectrum for the Duffing model.

In this case, there is a spurious Lyapunov exponent which converges to zero. It corresponds to the additional trivial evolution equation $\dot{t} = 1$.

LCEs of Discrete Systems

The discrete-time case is much simpler because, by the chain rule,

$$D_x f^t(x_0) = J(f^{t-1}(x_0)) \cdots J(f(x_0)) \cdot J(x_0),$$

where $J(x_0) = D_x f(x)\,\big|_{x=x_0}$. To avoid the above-mentioned numerical problems in the calculation of $D_x f^t(x)$, and to show a different approach to the numerical calculation of the whole LCE spectrum, here we prefer to follow the simpler algorithm of [Eckmann and Ruelle 1985]. We start by using the classical QR decomposition to write $J(x)$ as $J(x) = Q_1 R_1$, where Q_1 is an orthogonal matrix and R_1 is upper triangular. Then, for $k = 2, 3, \ldots, t$, we define

$$J_k^* = J(f^{k-1}(x))\, Q_{k-1} \qquad (9)$$

and decompose $J_k^* = Q_k R_k$. Clearly, $D_x f^t(x) = Q_t R_t \cdots R_1$. It is possible to show that the diagonal elements $v_{ii}^{(t)}$ of the upper-triangular matrix product $\Upsilon^{(t)} = R_t \cdots R_1$ satisfy

$$\lim_{t \to \infty} \frac{1}{t} \ln v_{ii}^{(t)} = \lambda_i.$$

Let us take, for example, the four-dimensional discrete-time dynamical system

```
In[35]:= G[{x_, y_, z_, w_}] := {1.85 - z^2 - 0.05 w, x, y, z}
```

We can estimate the Lyapunov spectrum in the following way. First, define the starting condition and the system dimension:

```
In[36]:= x0 = {0.1, 0, 0, 0};

In[37]:= n = Length[x0];
```

Then, compute the QR decomposition of the Jacobian $J(x)$, after a 100-step transient:

```
In[38]:= x = Array[a, n];

In[39]:= J[y_] := JacobianMatrix[G[x], x] /. Thread[x -> y]

In[40]:= xt = Nest[G, N[x0], 100];

In[41]:= {q, r} = QRDecomposition[J[xt]];
```

Finally, repeat K times the QR decomposition procedure (9):

```
In[42]:= K = 1000;

In[43]:= s={};

In[44]:= Do[
           xt = G[xt];
           {q, r} = QRDecomposition[J[xt].Transpose[q]];
           s = Append[s, Table[r[[i,i]], {i, n}]],
           {K}];

In[45]:= lces = Rest[FoldList[Plus, 0, Re[Log[s]]]]/Range[K];
```

The LCEs are

```
In[46]:= Last[lces]

Out[46]= {0.16985, 0.160085, 0.156375, -3.48204}

In[47]:= ConvergencePlot[lces]
```



FIGURE 5. Convergence plot of the Lyapunov spectrum for the hyper^2 chaotic map.

Again, we can obtain the same results using the LCEsD command defined in LCE.m:

```
In[48]:= lceshyper2 = LCEsD[G, {0.1, 0, 0, 0}, 1000, 100]

Out[48]= {{0.16985, 0.160085, 0.156375, -3.48204}, 3.13966}
```

The presence of three positive LCEs points out that the orbits of the system lie on a so-called "hyper^2-chaotic attractor" [Klein and Baier 1991].

Lyapunov Dimension

Kaplan and Yorke [1979, 228] have suggested an interesting conjecture that relates the fractal dimension of the attractor to the Lyapunov spectrum:

$$D_L = j - \frac{\sum_{i=1}^{j} \lambda_i}{\lambda_{j+1}},$$

where the LCEs are ordered in the usual way as $\lambda_1 \geq \ldots \geq \lambda_n$ and where j is the largest integer such that $\lambda_1 + \cdots + \lambda_j > 0$. In particular, Kaplan and Yorke suggest that D_L is a lower bound of the capacity dimension, that is, $D_L \leq D_C$. For more details, see [Farmer et al. 1983].

```
In[49]:= LyapunovDimension[x_] :=
         Module[{l, suml, j},
           l = Sort[x, Greater];
           suml = Rest[FoldList[Plus, 0, l]];
           j = Last[Position[suml, _?Positive]];
           First[j - suml[[j]]/l[[j+1]]] ]
```

For the Rössler hyperchaos example, the Lyapunov dimension is given by

```
In[50]:= LyapunovDimension[First[lcesrossler]]

Out[50]= 3.00596
```

For the hyper^2-chaotic attractor, D_L is

```
In[51]:= LyapunovDimension[First[lceshyper2]]
Out[51]= 3.13966
```

Acknowledgments

I am very grateful to Prof. Giancarlo Benettin for his illuminating lectures on ergodic theory and many helpful discussions. Special thanks to Gianna Zuliani and Elena Alberti for their kind suggestions, and Giovanni Tondini for placing *Mathematica* at my disposal. I am also grateful to Troels Petersen and an anonymous reviewer for useful revisions to the paper and associated *Mathematica* program.

References

Beltrami, E. 1989. *Mathematics for Dynamic Modeling.* Boston: Academic Press.

Benettin, G., L. Galgani, A. Giorgilli, and J.M. Strelcyn. 1980. Lyapunov characteristic exponents for smooth dynamical systems and for Hamiltonian systems: A method for computing all of them. Part 2: Numerical application. *Meccanica* 15:21–30.

Eckmann, J.-P., and D. Ruelle. 1985. Ergodic theory of chaos and strange attractors. *Reviews of Modern Physics* 57 (3, Part I): 617–656.

Falconer, K.J. 1985. *The Geometry of Fractal Sets.* Cambridge University Press.

Farmer, J.D., E. Ott, and J.A. Yorke. 1983. The dimension of chaotic attractors. *Physica* 7D:153–180.

Geist, K., U. Parlitz, and W. Lauterborn. 1990. Comparison of different methods for computing Lyapunov exponents. *Progress of Theoretical Physics* 83(5): 875–893.

Guckenheimer, J., and P. Holmes. 1983. *Nonlinear Oscillations, Dynamical Systems and Bifurcation of Vector Fields.* Berlin: Springer-Verlag.

Haken, H. 1983. At least one Lyapunov exponent vanishes if the trajectory of an attractor does not contain a fixed point. *Physics Letters* 94A(2): 71–72.

Kaplan, J., and J.A. Yorke. 1979. Functional differential equations and the approximation of fixed points. In *Summerschool and Conference on Functional Differential Equations and Approximation of Fixed Points*, University of Bonn, 1978, edited by H.-O. Peitgen and H.-O. Walther. Lectures Notes in Mathematics 730. Springer-Verlag.

Klein, M., and G. Baier. 1991. Hierachies of dynamical systems. In *A Chaotic Hierarchy*, edited by G. Baier and M. Klein. Singapore: World Scientific.

Maeder, R. 1991. *Programming in Mathematica.* 2nd ed. Addison-Wesley.

Oseledec, V.I. 1968. A multiplicative ergodic theorem: Lyapunov characteristic numbers for dynamical systems. *Trans. Moscow Math. Soc.* 19:197–231.

Parker, T.S., and L.O. Chua. 1989. *Practical Numerical Algorithms for Chaotic Systems.* New York: Springer-Verlag.

Rössler, O.E. 1991. The chaotic hierarchy. In *A Chaotic Hierarchy*, edited by G. Baier and M. Klein. Singapore: World Scientific.

Wiggins, S. 1988. *Global Bifurcations and Chaos: Analytical Methods.* New York: Springer-Verlag.

Marco Sandri
Università degli Studi di Verona
Istituto di Scienze Economiche, Via dell'Artigliere 19
I-37129 Verona, Italy
msandri@chiostro.univr.it

The electronic supplement contains the package `LCE.m`.